
Overview

This document lists TTCN changes needed to correct a problem in the ‘iwd-EUTRA-B2009-12_D11wk04’.

Corrections 
Change 1 / Correction of ‘Dsr_TransMax’ settings 

Function name
f_TC_7_3_5_2_EUTRA ( )

Reason for change
It is not possible to perform data sending, power change and Inter-cell HO within 60 ms timeframe, which is currently the case using default ‘Dsr_TransMax’ settings.

Summary of change
‘Dsr_TransMax’ is configured. Note that the proposed change is partly overlapping with previously raised R5s110033.

TTCN module
PDCP_Testcases.ttcn 

MCC160 Comment


Before change:

function f_TC_7_3_5_2_EUTRA() runs on EUTRA_PTC
{ // @sic R5s100305 sic@ @sic R5-104794 sic@
  var PDCP_SS_State_Type v_PDCP_Rec;
  var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info;
  // Initialization, generic setup, DRB loopback
  f_EUTRA_Init(c1);
  // Use 12 bits SN length
  fl_InitPDCP_Record(v_PDCP_Rec, AM_Mode, PDCP_SNLength12);
  // get FDD_TDD_Info
  v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);
  f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell2, -79 );
  // Cell configuration, with one AM DRB at the SS side
  f_EUTRA_CellConfig_Def (eutra_Cell1);
  // Setup cell 2
  f_EUTRA_CellInfo_SetC_RNTI ( eutra_Cell2, tsc_C_RNTI_Def3 );
  f_EUTRA_CellConfig_Def( eutra_Cell2 );
  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)
  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);
  //set null AS ciphering algorithm
  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);
  //entering UE test state 3, default values with BSR and PHR disabled
  f_EUTRA_RbEst_Def_DisableBSR_PHR ( eutra_Cell1 );
  // reconfigure DRB locally to set the required PDCP test mode
  f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB ( eutra_Cell1, cs_DRB1_ConfigAM_PDCP_Mode );
  // closing the test loop (mode A), entering state 4
  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling );
  // configure PDCP test mode for Cell 2
  f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB ( eutra_Cell2, cs_DRB1_ConfigAM_PDCP_Mode );
  f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@
  f_EUTRA_7_3_5_2_TestBody (v_PDCP_Rec);
  f_EUTRA_TestBody_Set(false);
  // leaving UE test state 4
  //@sic R5-103847 Deactivate test loop and open test loop is removed sic@
  //postamble
  f_EUTRA_Postamble (eutra_Cell2, E3_TESTMODE);
} // f_TC_7_3_5_2_EUTRA


After change:

function f_TC_7_3_5_2_EUTRA() runs on EUTRA_PTC
{ // @sic R5s100305 sic@ @sic R5-104794 sic@
  var PDCP_SS_State_Type v_PDCP_Rec;
  var template (value) EUTRA_FDD_TDD_CellInfo_Type  v_EUTRA_FDD_TDD_Info;
  var template (value) PhysicalConfigDedicated v_PhysicalConfigDedicated;
  // Initialization, generic setup, DRB loopback
  f_EUTRA_Init(c1);
  // Use 12 bits SN length
  fl_InitPDCP_Record(v_PDCP_Rec, AM_Mode, PDCP_SNLength12);
  // get FDD_TDD_Info
  v_EUTRA_FDD_TDD_Info:=f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);
  f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell2, -79 );
  // Cell configuration, with one AM DRB at the SS side
  f_EUTRA_CellConfig_Def (eutra_Cell1);
  // Setup cell 2
  f_EUTRA_CellInfo_SetC_RNTI ( eutra_Cell2, tsc_C_RNTI_Def3 );
  f_EUTRA_CellConfig_Def( eutra_Cell2 );
  // Preamble to enter UE in E-UTRA RRC_IDLE (state 2)
  f_EUTRA_Preamble (eutra_Cell1, STATE2A_TESTLOOP_ModeA);
  //set null AS ciphering algorithm
  f_EUTRA_AS_CipheringAlgorithm_Set(eea0);
  //entering UE test state 3, default values with BSR and PHR disabled
  // f_EUTRA_RbEst_Def_DisableBSR_PHR ( eutra_Cell1 );
v_PhysicalConfigDedicated := f_PhysicalConfigDedicated_Dsr_TransMax( eutra_Cell1 );

  f_EUTRA_GenericRbEst_DisableBSR_PHR ( eutra_Cell1, 0, 0, omit, omit, v_PhysicalConfigDedicated, omit, omit);
  // reconfigure DRB locally to set the required PDCP test mode
  f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB ( eutra_Cell1, cs_DRB1_ConfigAM_PDCP_Mode );
  // closing the test loop (mode A), entering state 4
  f_EUTRA_CloseUE_TestLoopModeA ( eutra_Cell1, cs_UE_TestLoopModeA_LB_Setup_NoScaling );
  // configure PDCP test mode for Cell 2
  f_EUTRA_SS_CommonRadioBearerConfig_SingleDRB ( eutra_Cell2, cs_DRB1_ConfigAM_PDCP_Mode );
  f_EUTRA_TestBody_Set(true, layer2);  // @sic R5-100050 sic@
  f_EUTRA_7_3_5_2_TestBody (v_PDCP_Rec);
  f_EUTRA_TestBody_Set(false);
  // leaving UE test state 4
  //@sic R5-103847 Deactivate test loop and open test loop is removed sic@
  //postamble
  f_EUTRA_Postamble (eutra_Cell2, E3_TESTMODE);
} // f_TC_7_3_5_2_EUTRA


Change 2 / Correction of UL grant configuration in TestBody 

Function name
f_EUTRA_7_3_5_2_TestBody (...)

Reason for change
UL grant configuration is wrong for HO, since PDCP Status Report is expected first separately after HO.

Summary of change
UL grant is corrected for HO, default grant is started after PDCP Status Reporting.

TTCN module
PDCP_Testcases.ttcn 

MCC160 Comment


Before change:

function f_EUTRA_7_3_5_2_TestBody(inout PDCP_SS_State_Type p_PDCP_Rec) runs on EUTRA_PTC
...

  f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_Cell1,  // @sic R5s100430 change 2; R5s100524; R5s110033 sic@
                                         eutra_Cell2,
                                         { cs_DRB1_ConfigAM_PDCP_Mode },
                                         tsc_C_RNTI_Def3, -, -, cs_UL_GrantConfig_Stop);
  //@siclog "Step 8A" siclog@
  // The UE sends DCP Control PDU via RLC-AM RB with the following content to the SS:
  //          D/C field  = 0 (PDCP control PDU) and PDU Type =000, FMS field = 5.
  DRB.receive (car_DRB_PDCP_PDU_Ind( eutra_Cell2,
                                     p_PDCP_Rec.RB_Id,
                                     cr_PDCP_StatusRep( int2bit(5, tsc_PDCP_SN_Size_12), omit )));
  //@siclog "Step 8B" siclog@
  // The SS generates a PDCP status report message and sends it to UE: D/C field = 0 (PDCP control PDU) and PDU Type =000, FMS field = 2.
  DRB.send (cas_DRB_PDCP_PDU_Req( eutra_Cell2,
                                  cs_TimingInfo_Now,
                                  p_PDCP_Rec.RB_Id,
                                  cs_PDCP_StatusRep ( int2bit(2,tsc_PDCP_SN_Size_12), omit )));
  // EXCEPTION: Step 9 shall be repeated for m=2 to 4 (increment=1).
  //@siclog "Step 9" siclog@
  // Check: Does the UE send the PDCP Data PDU #m via RLC-AM RB with the following content to the SS:
  // D/C field = 1 (PDCP Data PDU) and PDCP SN = m?
  // Data is previously received data from PDU #m.
...

} // f_EUTRA_7_3_5_2_TestBody


After change:

function f_EUTRA_7_3_5_2_TestBody(inout PDCP_SS_State_Type p_PDCP_Rec) runs on EUTRA_PTC
...

  f_EUTRA_508RRC_IntraLTE_HO_InterCell ( eutra_Cell1,  // @sic R5s100430 change 2; R5s100524; R5s110033 sic@
                                         eutra_Cell2,
                                         { cs_DRB1_ConfigAM_PDCP_Mode },
                                         tsc_C_RNTI_Def3, -,                                                                                                             cs_DciInfo_CcchDcchDtchUL_Explicit(2,2), cs_UL_GrantConfig_Stop, cs_UL_GrantConfig_OneTime);
  //@siclog "Step 8A" siclog@
  // The UE sends DCP Control PDU via RLC-AM RB with the following content to the SS:
  //          D/C field  = 0 (PDCP control PDU) and PDU Type =000, FMS field = 5.
  DRB.receive (car_DRB_PDCP_PDU_Ind( eutra_Cell2,
                                     p_PDCP_Rec.RB_Id,
                                     cr_PDCP_StatusRep( int2bit(5, tsc_PDCP_SN_Size_12), omit )));
  //@siclog "Step 8B" siclog@
  // The SS generates a PDCP status report message and sends it to UE: D/C field = 0 (PDCP control PDU) and PDU Type =000, FMS field = 2.
  DRB.send (cas_DRB_PDCP_PDU_Req( eutra_Cell2,
                                  cs_TimingInfo_Now,
                                  p_PDCP_Rec.RB_Id,
                                  cs_PDCP_StatusRep ( int2bit(2,tsc_PDCP_SN_Size_12), omit )));
f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now);
// EXCEPTION: Step 9 shall be repeated for m=2 to 4 (increment=1).
  //@siclog "Step 9" siclog@
  // Check: Does the UE send the PDCP Data PDU #m via RLC-AM RB with the following content to the SS:
  // D/C field = 1 (PDCP Data PDU) and PDCP SN = m?
  // Data is previously received data from PDU #m.
...

} // f_EUTRA_7_3_5_2_TestBody


Change 3 / Correction of group functions & templates for handling all expected PDCP PDUs in one TTI 

Function / template name
group PDCP_Functions
group PDCP_Templates

Reason for change
Received data handling is wrong after HO, since the 3 expected PDCP PDUs are coming up in one TTI with default grant configuration (according to the prose), so they should be expected in one list.

Summary of change
PDCP PDUs are expected within one PDU list with 3 elements. This also includes new template and function definitions.

TTCN module
PDCP_Testcases.ttcn 

MCC160 Comment


Before change:

group PDCP_Functions {
…

} // f_UPRcv2PDUs
} // group PDCP_Functions

group PDCP_Templates {
//ASP Constraint
template DRB_COMMON_IND car_DRB1_PDCP_DataListLongSN_2(CellId_Type p_CellId,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN1,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN2) :=
{ /* @status    APPROVED */
Common := cr_IndAspCommonPart_RB (p_CellId, tsc_RbId_DRB1),
  U_Plane := {
    SubframeData := {
      PduSduList := {
        PdcpPdu := {
          {DataLongSN := p_PDCP_PDULongSN1},
          {DataLongSN := p_PDCP_PDULongSN2}
        }
      },
      NoOfTTIs := ?
    }
  } // U_Plane
}; // car_DRB1_PDCP_DataListLongSN_2
template (value) PDCP_PDU_Type cs_PDCP_StatusRep ( B12_Type p_FMS,
                                                   template (omit) octetstring p_Bitmap) :=
{ /* @status    APPROVED */
  StatusReport := {
    D_C := tsc_D_C_CtrlPDU,
    Type := tsc_PDU_Type_StatusRep,
    FMS := p_FMS,
    Bitmap := p_Bitmap
  }
};
…

  // EXCEPTION: Step 9 shall be repeated for m=2 to 4 (increment=1).
  //@siclog "Step 9" siclog@
  // Check: Does the UE send the PDCP Data PDU #m via RLC-AM RB with the following content to the SS:
  // D/C field = 1 (PDCP Data PDU) and PDCP SN = m?
  // Data is previously received data from PDU #m.
  f_UPRcv (p_PDCP_Rec, eutra_Cell2, v_Data3);
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");
  f_UPRcv (p_PDCP_Rec, eutra_Cell2, v_Data4);
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");
  f_UPRcv (p_PDCP_Rec, eutra_Cell2, v_Data5);
  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");
} // f_EUTRA_7_3_5_2_TestBody


After change:

group PDCP_Functions {
…

} // f_UPRcv2PDUs
function f_UPRcv3PDUs (inout PDCP_SS_State_Type p_PDCP_Rec,
                       CellId_Type p_CellId,
                       template (present) octetstring p_PdcpSdu1,
                       template (present) octetstring p_PdcpSdu2,
                       template (present) octetstring p_PdcpSdu3)
  runs on EUTRA_PTC
{
  var integer v_Next_PDCP_RX_SN := p_PDCP_Rec.Next_PDCP_RX_SN;
  var PDCP_SDU_Type v_ReceivedSdu1;
  var PDCP_SDU_Type v_ReceivedSdu2;
  var PDCP_SDU_Type v_ReceivedSdu3;
  var DRB_COMMON_IND v_ReceivedPdu;
  // for the time being RX_HFN will always be 0, so Next_PDCP_RX_SN is used instead of RX_COUNT !!!
  // Build SN template to be used
  if (p_PDCP_Rec.PDCP_SN_Size != PDCP_SNLength12)
    {
      FatalError(__FILE__, __LINE__, "PDCP_SN_Size incorrect");
    }
  DRB.receive (car_DRB1_PDCP_DataListLongSN_3(p_CellId,
                                              cr_PDCP_DataPdu_LongSN(int2bit(v_Next_PDCP_RX_SN, 12), p_PdcpSdu1),
                                              cr_PDCP_DataPdu_LongSN(int2bit((v_Next_PDCP_RX_SN + 1), 12), p_PdcpSdu2),
                                              cr_PDCP_DataPdu_LongSN(int2bit((v_Next_PDCP_RX_SN + 2), 12), p_PdcpSdu3)))-> value v_ReceivedPdu;
  v_ReceivedSdu1 := v_ReceivedPdu.U_Plane.SubframeData.PduSduList.PdcpPdu[0].DataLongSN.SDU;
  v_ReceivedSdu2 := v_ReceivedPdu.U_Plane.SubframeData.PduSduList.PdcpPdu[1].DataLongSN.SDU;
  v_ReceivedSdu3 := v_ReceivedPdu.U_Plane.SubframeData.PduSduList.PdcpPdu[2].DataLongSN.SDU;
  fl_DecipherAndMatchReceivedSDU(p_PDCP_Rec, v_ReceivedSdu1, p_PdcpSdu1);
  fl_DecipherAndMatchReceivedSDU(p_PDCP_Rec, v_ReceivedSdu2, p_PdcpSdu2);
  fl_DecipherAndMatchReceivedSDU(p_PDCP_Rec, v_ReceivedSdu3, p_PdcpSdu3);
} // f_UPRcv3PDUs
} // group PDCP_Functions

group PDCP_Templates {
//ASP Constraint
template DRB_COMMON_IND car_DRB1_PDCP_DataListLongSN_2(CellId_Type p_CellId,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN1,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN2) :=
{ /* @status    APPROVED */
Common := cr_IndAspCommonPart_RB (p_CellId, tsc_RbId_DRB1),
  U_Plane := {
    SubframeData := {
      PduSduList := {
        PdcpPdu := {
          {DataLongSN := p_PDCP_PDULongSN1},
          {DataLongSN := p_PDCP_PDULongSN2}
        }
      },
      NoOfTTIs := ?
    }
  } // U_Plane
}; // car_DRB1_PDCP_DataListLongSN_2
template DRB_COMMON_IND car_DRB1_PDCP_DataListLongSN_3(CellId_Type p_CellId,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN1,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN2,
                                                       template (present) PDCP_DataPdu_LongSN_Type p_PDCP_PDULongSN3) :=
{ /* @status    APPROVED */
Common := cr_IndAspCommonPart_RB (p_CellId, tsc_RbId_DRB1),
  U_Plane := {
    SubframeData := {
      PduSduList := {
        PdcpPdu := {
          {DataLongSN := p_PDCP_PDULongSN1},
          {DataLongSN := p_PDCP_PDULongSN2},
          {DataLongSN := p_PDCP_PDULongSN3}
        }
      },
      NoOfTTIs := ?
    }
  } // U_Plane
}; // car_DRB1_PDCP_DataListLongSN_3
template (value) PDCP_PDU_Type cs_PDCP_StatusRep ( B12_Type p_FMS,
                                                   template (omit) octetstring p_Bitmap) :=
{ /* @status    APPROVED */
  StatusReport := {
    D_C := tsc_D_C_CtrlPDU,
    Type := tsc_PDU_Type_StatusRep,
    FMS := p_FMS,
    Bitmap := p_Bitmap
  }
};
…

  // EXCEPTION: Step 9 shall be repeated for m=2 to 4 (increment=1).
  //@siclog "Step 9" siclog@
  // Check: Does the UE send the PDCP Data PDU #m via RLC-AM RB with the following content to the SS:
  // D/C field = 1 (PDCP Data PDU) and PDCP SN = m?
  // Data is previously received data from PDU #m.
  f_UPRcv3PDUs (p_PDCP_Rec, eutra_Cell2, v_Data3, v_Data4, v_Data5);
} // f_EUTRA_7_3_5_2_TestBody

